

# Structural Logical Relations\*

Carsten Schürmann  
IT University of Copenhagen  
Copenhagen, Denmark  
carsten@itu.dk

Jeffrey Sarnat  
Yale University  
New Haven, CT, USA  
sarnat@cs.yale.edu

## Abstract

*Tait’s method (a.k.a. proof by logical relations) is a powerful proof technique frequently used for showing foundational properties of languages based on typed  $\lambda$ -calculi. Historically, these proofs have been extremely difficult to formalize in proof assistants with weak meta-logics, such as Twelf, and yet they are often straightforward in proof assistants with stronger meta-logics. In this paper, we propose structural logical relations as a technique for conducting these proofs in systems with limited meta-logical strength by explicitly representing and reasoning about an auxiliary logic. In support of our claims, we give a Twelf-checked proof of the completeness of an algorithm for checking equality of simply typed  $\lambda$ -terms.*

## 1. Introduction

When formalizing meta-theoretic proofs about formal systems, one is usually first confronted with choosing a proof assistant. This decision is usually driven by concerns regarding the expressive power of the proof assistant, the trust in its logical foundation, performance, the features of the tactic language, and the ease with which we can encode the formal system.

For example, Thomas Hales has chosen HOL Light [Har96] for his work on the formalization of Kepler’s conjecture. George Necula has chosen variants of the LF type theory for his original work on proof-carrying code [Nec97]. Daniel Lee, Karl Crary, and Bob Harper have chosen the Twelf system for formalizing the proof of the soundness of an SML type system [LCH07] and Georges Gonthier has chosen Coq for his formalized proof of the four color theorem [Gon05]. The solutions posted to the POPLmark challenge employ a whole range of different proof assistants, including Twelf, Coq, Isabelle/HOL, Matita, Alpha Prolog, and ATS.

The proof theoretic strength of the different systems vary significantly. Once the proof assistant question is settled and a proof development has begun, there is often no turning back. If unforeseen challenges in the formalization of a meta-theoretic argument arise, one can find oneself in a situation where one is seemingly stuck. This might suggest that the best proof assistant is necessarily the most expressive, but such a perspective is too naive with respect to other considerations that influence the choice of proof assistant.

In this paper we present a technique called *structural logical relations* that avails proofs by logical relations to systems with limited meta-logical strength by explicitly representing and reasoning about an auxiliary logic. Proofs by structural logical relations follow the same line of reasoning as their informal counterparts. The central idea is to formalize the relevant part of the argument in this auxiliary logic, which we refer to as the *assertion logic*. Structural logical relations are therefore best employed in proof assistants based on logical frameworks, such as Twelf, that provide adequate, elegant, and convenient to work with encodings of proofs in the assertion logic.

We show how to formalize structural logical relations in practice using the meta-logical framework Twelf. But this paper is not just about the *how*, it also highlights some insights into the proof itself. For example, we illustrate that a proof by logical relations can be viewed as a recipe to reduce the provability of a theorem to the normalizability of proofs in the assertion logic. We present the results in such a way that the reader interested in the nature of logical proofs, but not in their formalization, should be able to follow this paper by ignoring the formalized material.

To best follow the paper, however, it is important that we distinguish carefully between the *logical framework*, in our case LF [HHP93] that serves the formalization of judgments, the *assertion logic* that we will encode in LF, and the *meta-logic* of Twelf, in which we carry out proofs by structural logical relations. Our proofs are constructive and executable, laying to rest the common misconception that the meta-logical framework Twelf is unable to reason with proofs by logical relations. The  $FOL^{\Delta N}$

---

\*This research has been funded by the National Science Foundation under grant CCR-0325808.

system [MM97] has a similar distinction between logical framework and meta-logic; we suspect that structural logical relations would be convenient in that setting as well.

This paper is structured as follows: In Section 2 we define the term *structural logical relation* and give a simple yet useful example of an assertion logic, namely minimal first-order logic, suitable for reasoning about simply typed languages. In Section 3, we briefly comment on the formalization of the technique in the proof assistant Twelf. We give an introductory example of a structural logical relations proof, namely the weak normalization for the simply typed  $\lambda$ -calculus in Section 4 before we address the goal of this paper, the completeness of equivalence checking in Section 5. An identifying feature of our structural logical relations proofs is the cut-elimination property of the assertion logic. Therefore, we study the features and implications of cut-elimination of the assertion logic in Section 6. We conclude with Section 7 where we discuss implications and possible extensions. The accompanying Twelf source code [SS08] is available at <http://www.twelf.org/slr>.

## 2. Structural Logical Relations

A proof by logical relations (also sometimes referred to as Tait’s method) relies on an interpretation of types as relations between objects. On paper, these interpretations are usually formulated in set theory. In the spirit of intuitionism, we not only want to know that a certain property holds, but we would like to be able to produce a constructive witness of this fact.

Consider the simply typed  $\lambda$ -calculus.

$$\begin{array}{l} \text{Types} \quad \tau ::= \circ \mid \tau_1 \Rightarrow \tau_2 \\ \text{Expressions} \quad e ::= x^\tau \mid \text{lam } x^\tau. e \mid \text{app } e_1 e_2 \end{array}$$

We refer to a property that is logic independent as a *judgment*. Examples of judgments  $J$  include, for example,  $e$  reduces to a normal form,  $e$  weak-head reduces to  $e'$ , or  $e$  is  $\beta\eta$ -equivalent to  $e'$ . Given some judgment  $J(e)$ , a unary logical relation  $\llbracket \tau \rrbracket$  can be defined as the smallest set that satisfies the following two conditions.

$$\begin{array}{l} e \in \llbracket \circ \rrbracket \quad \text{iff} \quad J(e) \\ e \in \llbracket \tau_2 \Rightarrow \tau_1 \rrbracket \quad \text{iff} \quad \text{for all } e_2, \text{ if } e_2 \in \llbracket \tau_2 \rrbracket \\ \quad \quad \quad \text{then app } e e_2 \in \llbracket \tau_1 \rrbracket \end{array}$$

The hallmark characteristic of a logical relation is the definition at function types: functions are related if they map related arguments to related results. A proof by logical relations typically proceeds in two stages: first, it is shown that  $e \in \llbracket \tau \rrbracket$  implies  $J(e)$ , then it is shown that, for every  $e$  of type  $\tau$ ,  $e \in \llbracket \tau \rrbracket$ . A paper proof for the weak normalization of the simply typed  $\lambda$ -calculus can be found in [Pfe92]; a

proof using structural logical relations for the same property is given in Section 4.

Taking a step back, we notice that the logical relation defined above relies on two things: structural induction on  $\tau$ , judgments of the form of  $J$ , and connectives “for all” and “if ... then.” Logical relations are typically formulated with the understanding that the induction principle, judgments, and connectives live on the same level. The idea of *structural logical relations* is to separate these concepts: induction lives on the meta-level, whereas the judgments and connectives live elsewhere. To this end, we declare a logic, the *assertion logic*, that is expressive enough to define both a logical relation in terms of logical connectives and atomic formulas  $P$  that represent the judgments  $J$  in the logic. The formulas of our assertion logic that we use throughout this paper are defined by the following.

$$\begin{array}{l} \text{Formulas} \quad F, G ::= P \mid F \supset G \mid \forall x^\tau. F \\ \text{Assumptions} \quad \Delta ::= \cdot \mid \Delta, F \mid \Delta, x^\tau \end{array}$$

In the definition of expressions, every variable carries its own type, which means that we can restrict the universal quantifier to range over well-typed expressions even if these contain free variables. We write  $e^\tau$  for an expression  $e$  of type  $\tau$  and omit the superscript  $\tau$  whenever it can be easily inferred.

We define the judgment of derivability in a sequent calculus  $\Delta \vdash F$  by the following rules. If  $\Delta$  is  $\cdot$ , then we simply drop it from the judgment.

$$\begin{array}{c} \frac{F \in \Delta}{\Delta \vdash F} \text{ax} \quad \frac{\Delta, F \vdash G}{\Delta \vdash F \supset G} \text{impR} \\ \frac{\Delta, F \supset G \vdash F \quad \Delta, F \supset G, G \vdash H}{\Delta, F \supset G \vdash H} \text{impL} \\ \frac{\Delta, x^\tau \vdash F}{\Delta \vdash \forall x^\tau. F} \text{allR} \quad \frac{\Delta, \forall x^\tau. F, [e^\tau/x]F \vdash H}{\Delta, \forall x^\tau. F \vdash H} \text{allL} \end{array}$$

The rules are standard. We let the universal quantifier range over expressions  $e^\tau$ . We write  $[e^\tau/x]F$  for the usual notion of capture-avoiding substitution.

For practical reasons we will work with the assertion logic with cut. The judgment  $\Delta \vdash^{\text{cut}} F$  is defined by inference rules similar to the ones above except with  $\vdash^{\text{cut}}$  instead of  $\vdash$ , plus one additional rule.

$$\frac{\Delta \vdash^{\text{cut}} F \quad \Delta, F \vdash^{\text{cut}} G}{\Delta \vdash^{\text{cut}} G} \text{cut}$$

The rules for defining the atomic formulas  $P$  vary for each structural logical relation. Hence they are defined in Sections 4 and 5, where we discuss concrete applications of our technique. We show later in the paper that the choice of

inference rules cannot be arbitrary: they must not destroy the cut-elimination property of the sequent-calculus.

Returning to the definition of *structural logical relations*, we can now define  $\llbracket \tau \rrbracket$  more formally by meta-level induction on  $\tau$ :

$$\begin{aligned} \llbracket \circ \rrbracket(e) &= P(e) \\ \llbracket \tau_2 \Rightarrow \tau_1 \rrbracket(e) &= \forall e'^{\tau_2}. \llbracket \tau_2 \rrbracket(e') \supset \llbracket \tau_1 \rrbracket(\text{app } e \ e') \end{aligned}$$

And more generally, the definition of an  $n$ -ary structural logic relation is given by the following formula:

$$\begin{aligned} \llbracket \circ \rrbracket(e_1 \dots e_n) &= P(e_1 \dots e_n) \\ \llbracket \tau_2 \Rightarrow \tau_1 \rrbracket(e_1 \dots e_n) &= \forall e_1'^{\tau_2}. \dots \forall e_n'^{\tau_2}. \llbracket \tau_2 \rrbracket(e_1' \dots e_n') \\ &\quad \supset \llbracket \tau_1 \rrbracket(\text{app } e_1 \ e_1' \dots \text{app } e_n \ e_n') \end{aligned}$$

An example of a binary structural logical relation can be found in Section 5. Next, we derive the proof obligations justifying a proof by structural logical relations in the unary case.

1. If  $e^\tau$  then  $\vdash^{\text{cut}} \llbracket \tau \rrbracket(e)$ . (Fundamental Theorem)
2.  $\vdash^{\text{cut}} \llbracket \tau \rrbracket(e) \supset P(e)$ . (Escape Lemma)
3. If  $\vdash^{\text{cut}} P(e)$  then  $\vdash P(e)$ . (Cut-Elimination)
4. If  $\vdash P(e)$  then  $J(e)$ . (Extraction Theorem)

Note that, instead of defining the structural logical relation using meta-level induction, we could have chosen to extend the assertion logic axiomatically by an induction principle over types. However, this would have not only been less convenient, but would have also complicated the cut-elimination proof for our assertion logic, which we discuss in Sections 4 and 5.

### 3. The Meta-Logical Framework Twelf

Turning to the formalization of a structural logical relations argument, we follow [Pfe95], and describe adequate encodings of the simply typed  $\lambda$ -calculus and the sequent calculus in the logical framework LF [HHP93] using Twelf notation. First we address the formalization of simple types.

```
tp : type.
o  : tp.
=> : tp -> tp -> tp. %infix right 10 ==>.
```

Any LF object made from arrows  $\Rightarrow$  and base types  $\circ$  corresponds to a well-formed simple type, and vice versa. As  $\Rightarrow$  is declared infix,  $\circ \Rightarrow \circ \Rightarrow \circ$  is the representation of a function type that expects two arguments. In Twelf, we write `type` for the kind of a type, and use curly braces  $\{t : \text{tp}\}$  for dependent types and dependent kinds.

We write  $\rightarrow$  (infix) if the dependency is vacuous. Furthermore, we write  $[t : \text{tp}]$  for  $\lambda$ -abstraction,  $t : \text{tp}$  for type ascription, and juxtaposition for application in the logical framework.

Next, we define the syntactic category of terms  $e^\tau$ . If  $\tau$  is represented by  $T : \text{tp}$ , then we write  $\text{tm } T$  for the LF type representing simply typed expressions  $e^\tau$ . The definition takes advantage of the fact that, in LF, type families may be indexed by LF objects.

```
tm  : tp -> type.
lam : (tm T1 -> tm T2) -> tm (T1 => T2).
app : tm (T2 => T1) -> tm T2 -> tm T1.
```

The type of `lam` illustrates our use of higher-order abstract syntax, which we use pervasively throughout this paper: variables of the simply typed  $\lambda$ -calculus and assumptions about the being well-typed are represented by variables of LF.

Formulas of the assertion logic are encoded as LF objects of type `form`. We defer the formalization of atomic formulas to the subsequent relevant sections in this paper.

```
form  : type.
forall : (tm T -> form) -> form.
==>   : form -> form -> form.
       %infix right 10 ==>.
```

Every sequent derivation of  $F_1, \dots, F_n \vdash G$  can be adequately represented in LF as an object of type `conc G` in a context  $u_1 : \text{hyp } F_1, \dots, u_n : \text{hyp } F_n$ . We distinguish two variants of the assertion logic, the one without the cut rule is depicted first.

```
hyp   : form -> type.
conc  : form -> type.
ax    : hyp F -> conc F.
impr  : (hyp F -> conc G)
       -> conc (F ==> G).
impl  : conc F -> (hyp G -> conc H)
       -> (hyp (F ==> G) -> conc H).
forallr : ({x:tm T} conc (F x))
       -> conc (forall F).
foralll : {E: tm T} (hyp (F E) -> conc H)
       -> (hyp (forall F) -> conc H).
```

Analogously, every derivation of  $F_1, \dots, F_n \vdash^{\text{cut}} G$  can be adequately represented in LF as an object of type `conc* G`. We give the encoding of the cut rule, the others are obtained from the former by copying and replacing `conc` by `conc*`.

```
conc* : form -> type.
cut   : {F: form} conc* F
       -> (hyp F -> conc* G)
       -> conc* G.
```



$$\begin{array}{c}
\frac{\Delta \vdash \text{ha}^{\tau_2 \Rightarrow \tau_1}(e_1) \quad \Delta \vdash \text{hc}^{\tau_2}(e_2)}{\Delta \vdash \text{ha}^{\tau_1}(\text{app } e_1 e_2)} \text{ha\_app} \\
\frac{\Delta, x^{\tau_1}, \text{ha}^{\tau_1}(x) \vdash \text{hc}^{\tau_2}(\text{app } e x)}{\Delta \vdash \text{hc}^{\tau_1 \Rightarrow \tau_2}(e)} \text{hc\_arr} \\
\frac{\Delta \vdash \text{wh}^\circ(e, e') \quad \Delta \vdash \text{hc}^\circ(e')}{\Delta \vdash \text{hc}^\circ(e)} \text{hc\_wh} \\
\frac{\Delta \vdash \text{ha}^\circ(e)}{\Delta \vdash \text{hc}^\circ(e)} \text{hc\_atm} \\
\frac{}{\Delta \vdash \text{wh}^\tau(\text{app } (\text{lam } x. e_1) e_2, e_1[e_2/x])} \text{wh\_beta} \\
\frac{\Delta \vdash \text{wh}^{\tau_2 \Rightarrow \tau_1}(e_1, e'_1)}{\Delta \vdash \text{wh}^{\tau_1}(\text{app } e_1 e_2, \text{app } e'_1 e_2)} \text{wh\_cong}
\end{array}$$

The formulation of the sequent calculus with cut is extended in the same fashion. We formalize this in LF as follows.

```

hc: tm T -> form.
ha: tm T -> form.
wh: tm T -> tm T -> form.

ha_app : conc (ha E1) -> conc (hc E2)
         -> conc (ha (app E1 E2)).
hc_arr : ({x: tm T1} hyp (ha x)
         -> conc (hc (app E x)))
         -> conc (hc E).
hc_wh  : {E:tm o} conc (wh E E')
         -> conc (hc E')
         -> conc (hc E).
hc_atm : {E: tm o} conc (ha E)
         -> conc (hc E).
wh_beta: conc (wh (app (lam E1) E2) (E1 E2)).
wh_cong: conc (wh E1 E1')
         -> conc (wh (app E1 E2) (app E1' E2)).

```

The rules extending `conc*` are analogous. The next lemma and theorem illustrate how we can extract the canonical form together with a reduction trace from the cut-free derivation of  $\vdash \text{hc}^\tau(e)$ . In Twelf, theorems are formalized as relations between universally and existentially quantified derivations. We give the relations as LF type families at the end of selected proofs. The role that each argument to the relation plays can always be read out of the informal statement of the theorem.

#### Lemma 4.1 (Congruence)

1. If  $e \longrightarrow^* e'$  then  $\text{lam } x^\tau. e \longrightarrow^* \text{lam } x^\tau. e'$ .
2. If  $e_1 \longrightarrow^* e'_1$  and  $e_2 \longrightarrow^* e'_2$  then  $\text{app } e_1 e_2 \longrightarrow^* \text{app } e'_1 e'_2$ .

3. If  $e \xrightarrow{\text{whr}} e'$  then  $e \longrightarrow e'$ .

**Proof:** By straightforward structural inductions.  $\square$

**Theorem 4.2 (Extraction)** *All derivations in the sequent calculus are assumed to be cut-free. Let  $\Delta = x_1^{\tau_1}, \text{ha}^{\tau_1}(x_1), \dots, x_n^{\tau_n}, \text{ha}^{\tau_n}(x_n)$ .*

1. For all derivations of  $\Delta \vdash \text{hc}^\tau(e)$  there exist derivations of

$$\begin{array}{c}
\frac{}{x_1 \Downarrow \tau_1} u_1 \quad \dots \quad \frac{}{x_n \Downarrow \tau_n} u_n \\
\vdots \quad \vdots \quad \vdots \\
v \Uparrow \tau
\end{array}$$

and  $e \longrightarrow^* v$ .

2. For all derivations of  $\Delta \vdash \text{ha}^\tau(e)$  there exist derivations of

$$\begin{array}{c}
\frac{}{x_1 \Downarrow \tau_1} u_1 \quad \dots \quad \frac{}{x_n \Downarrow \tau_n} u_n \\
\vdots \quad \vdots \quad \vdots \\
v \Downarrow \tau
\end{array}$$

and  $e \longrightarrow^* v$ .

3. For all derivations of  $\Delta \vdash \text{wh}^\tau(e, e')$  there exists a derivation of  $e \xrightarrow{\text{whr}} e'$ .

**Proof:** By mutual induction on the derivations of  $\Delta \vdash \text{hc}^\tau(e)$ ,  $\Delta \vdash \text{ha}^\tau(e)$ , and  $\Delta \vdash \text{wh}^\tau(e, e')$ . 1. follows from Lemma 4.1 (1), (3) and rules `trans`, `exp_eta`, and `n1`, `n2`. 2. follows from Lemma 4.1 (2) and rule `atm.app`. 3. follows from the application of rules `whr_beta`, `whr_cong`.

```

ext1 : conc (hc E)
      -> {V} can V -> reds E V -> type.
ext2 : conc (ha E)
      -> {V} atm V -> reds E V -> type.
ext3 : conc (wh E E') -> whr E V -> type.

```

$\square$

We define the logical relation as a predicate in the assertion logic, by instantiating  $P$  (from the general definition of a unary logical relation from Section 2) with `hco`.

#### Definition 4.3 (Logical relation)

$$\begin{aligned}
\llbracket \text{o} \rrbracket(e) &= \text{hc}^\circ(e) \\
\llbracket \tau_2 \Rightarrow \tau_1 \rrbracket(e) &= \forall e_2^{\tau_2}. \llbracket \tau_2 \rrbracket(e_2) \supset \llbracket \tau_1 \rrbracket(\text{app } e e_2)
\end{aligned}$$

```

lr : {T:tp} (tm T -> form) -> type.
lr1: lr o (hc: tm o -> form).
lr2: lr T2 F -> lr T1 G
     -> lr (T2 => T1) ([e: tm (T2 => T1)]
                    (forall [e_2:tm T2] (F e2)
                    ==> (G (app e e2))))

```

## 4.2. Meta-Theory

All proofs in the remainder of this section are inductive over the structure of types, typing derivations, and proofs in the assertion logics. No definition of simultaneous substitutions is needed. No extension of the concept of logical relation to simultaneous substitutions is needed. And finally, no Kripke-style explicit contexts are needed in the definition of the logical relation. This information is already implicitly contained within the assertion logic proofs.

**Lemma 4.4 (Escape)** 1. *For all types  $\tau$  and assumptions  $\Delta$ , there exists a derivation of  $\Delta \vdash^{\text{cut}} \forall e^\tau. \llbracket \tau \rrbracket(e) \supset \text{hc}^\tau(e)$ .*

2. *For all types  $\tau$  and assumptions  $\Delta$ , there exists a derivation of  $\Delta \vdash^{\text{cut}} \forall e^\tau. \text{ha}^\tau(e) \supset \llbracket \tau \rrbracket(e)$ .*

**Proof:** By mutual induction on  $\tau$ .

**Case:**  $\tau = \text{o}$ . Direct, by reasoning using Def. 4.3 and rules allR, impR, and ax.

**Case:**  $\tau = \tau_2 \Rightarrow \tau_1$ . By induction hypothesis, we obtain that for all  $\Delta'$

$$\Delta' \vdash^{\text{cut}} \forall e^{\tau_1}. \llbracket \tau_1 \rrbracket(e) \supset \text{hc}^{\tau_1}(e) \quad (1)$$

$$\Delta' \vdash^{\text{cut}} \forall e^{\tau_1}. \text{ha}^{\tau_1}(e) \supset \llbracket \tau_1 \rrbracket(e) \quad (2)$$

$$\Delta' \vdash^{\text{cut}} \forall e^{\tau_2}. \llbracket \tau_2 \rrbracket(e) \supset \text{hc}^{\tau_2}(e) \quad (3)$$

$$\Delta' \vdash^{\text{cut}} \forall e^{\tau_2}. \text{ha}^{\tau_2}(e) \supset \llbracket \tau_2 \rrbracket(e) \quad (4)$$

1. Follows by induction hypotheses (1) and (4), using Def. 4.3 and rules allR, impR, allL, implL, cut, and hc.arr.
2. Follows by induction hypotheses (2) and (3), using rules allR, impR, allL, cut, and ha\_app.

esc1:  $\{\text{T:tp}\} \{\text{F:(tm T)} \rightarrow \text{form}\} \text{lr T F}$   
 $\rightarrow (\text{conc*} (\text{forall [e] F e} \Rightarrow \text{hc e})) \rightarrow \text{type}.$

esc2:  $\{\text{T:tp}\} \{\text{F:(tm T)} \rightarrow \text{form}\} \text{lr T F}$   
 $\rightarrow (\text{conc*} (\text{forall [e] ha e} \Rightarrow \text{F e})) \rightarrow \text{type}.$

□

**Lemma 4.5 (Closure under weak head expansion)** *For all types  $\tau$  and assumptions  $\Delta$ , there exists a derivation of  $\Delta \vdash^{\text{cut}} \forall e^\tau. \forall e'^\tau. \text{wh}^\tau(e, e') \supset \llbracket \tau \rrbracket(e') \supset \llbracket \tau \rrbracket(e)$  in the assertion logic.*

**Proof:** By induction on  $\tau$ .

**Case:**  $\tau = \text{o}$ . Direct, by reasoning using Def. 4.3 and rules allR, impR, ax, and hc\_wh.

**Case:**  $\tau = \tau_2 \Rightarrow \tau_1$ . By induction hypothesis, we obtain that for all  $\Delta'$

$$\Delta' \vdash^{\text{cut}} \forall e^{\tau_1}. \forall e'^{\tau_1}. \text{wh}^{\tau_1}(e, e') \supset \llbracket \tau_1 \rrbracket(e') \supset \llbracket \tau_1 \rrbracket(e)$$

The claim follows from Def. 4.3 and rules allR, impR, allL, implL, cut, and wh\_cong.

cwhe:  $\{\text{T:tp}\} \{\text{F:(tm T)} \rightarrow \text{form}\} \text{lr T F}$   
 $\rightarrow (\text{conc*} (\text{forall [e] forall [e']}$   
 $\quad \text{wh e e' ==> F e' ==> F e}))$   
 $\rightarrow \text{type}.$

□

**Theorem 4.6 (Fundamental)** *For all types  $\tau$  and assumptions  $\Delta = x_1^{\tau_1}, \llbracket \tau_1 \rrbracket(x_1), \dots, x_n^{\tau_n}, \llbracket \tau_n \rrbracket(x_n)$  and for all  $e^\tau$  with free variables among  $x_1 \dots x_n$ , there exists a derivation of  $\Delta \vdash^{\text{cut}} \llbracket \tau \rrbracket(e)$ .*

**Proof:** By induction on the structure of  $e : \tau$ .

**Case:**  $\tau = \tau_i, e = x_i$ . Direct, by rule ax.

**Case:**  $e = \text{app } e_1 e_2$ . The claim follows from the induction hypotheses  $\Delta \vdash^{\text{cut}} \llbracket \tau \rrbracket(e_1)$  and  $\Delta \vdash^{\text{cut}} \llbracket \tau \rrbracket(e_2)$  using Def. 4.3 and rules ax, allL, implL.

**Case:**  $e = \text{lam } x^{\tau_{n+1}}. e'$ . The claim follows from the induction hypothesis  $\Delta, x_{n+1}^\tau, \llbracket \tau_{n+1} \rrbracket(x) \vdash^{\text{cut}} \llbracket \tau' \rrbracket(e')$  using Def. 4.3, Lemma 4.5, and rule wh\_beta.

fund:  $\{\text{E:tm T}\} \text{lr T F}$   
 $\rightarrow \text{conc*} (\text{F E}) \rightarrow \text{type}.$

□

In order to complete the proof, we need to appeal to a proof normalization technique for our assertion logic: cut-elimination. For any well-typed  $e^\tau$ , after applying the Fundamental Theorem 4.6, followed by the Escape Lemma 4.4, followed by a few applications of the rules cut, forallll, and impl, we obtain a *cut-full* proof of  $\vdash^{\text{cut}} \text{hc}^\tau(e)$ . The Extraction Theorem 4.2, however, requires the proof of  $\vdash \text{hc}^\tau(e)$  to be *cut-free*. It is this step where the true work of the structural logical relation proof takes place, as we need to eliminate all cuts in order to apply the Extraction Theorem.

**Theorem 4.7 (Cut-elimination)** *Let  $\Delta$  denote assumptions. Any cut-full sequent derivation of  $\Delta \vdash^{\text{cut}} F$  can be converted into  $\Delta \vdash F$ .*

ce :  $\text{conc*} F \rightarrow \text{conc} F \rightarrow \text{type}.$

**Proof:** See: Pfenning [Pfe95]. The extra cases introduced by our atomic propositions all fall into the category of “right commutative conversions.” □



$$\begin{array}{c}
\frac{e_1 \rightsquigarrow e'_1 \quad e_2 \rightsquigarrow e'_2 \quad e'_1 \sim e'_2 \downarrow \circ}{e_1 \sim e_2 \uparrow \circ} \text{qat\_base} \\
\frac{\quad}{x \sim x \downarrow \tau} u \\
\vdots \\
\frac{\text{app } e \ x \sim \text{app } e' \ x \uparrow \tau_2}{e \sim e' \uparrow \tau_1 \Rightarrow \tau_2} \text{qat\_arrow}^{x,u} \\
\frac{e_1 \sim e'_1 \downarrow \tau_2 \Rightarrow \tau_1 \quad e_2 \sim e'_2 \uparrow \tau_2}{\text{app } e_1 \ e_2 \sim \text{app } e'_1 \ e'_2 \downarrow \tau_1} \text{qap\_app}
\end{array}$$

We write

$\text{algC} : \text{tm } \Gamma \rightarrow \text{tm } \Gamma \rightarrow \text{type}.$

$\text{algA} : \text{tm } \Gamma \rightarrow \text{tm } \Gamma \rightarrow \text{type}.$

for  $e \sim e' \uparrow \tau$  and  $e \sim e' \downarrow \tau$ , respectively.

### 5.1. The Structural Logical Relation

We shall prove that if  $e \equiv e' : \tau$  then  $e \sim e' \uparrow \tau$ . Following the example in Section 4, we extend the assertion logic defined in Section 2 by binary versions of the predicates  $\text{ha}^\tau(e, e')$  and  $\text{hc}^\tau(e, e')$  at types  $\tau$ , and the corresponding inference rules. As for weak-head reduction, we simply reuse the definition of the binary predicate  $\text{wh}^\tau(e, e')$  and specify the meaning of the new predicates by the following rules.

$$\begin{array}{c}
\frac{\Delta \vdash \text{ha}^{\tau_2 \Rightarrow \tau_1}(e_1, e'_1) \quad \Delta \vdash \text{hc}^{\tau_2}(e_2, e'_2)}{\Delta \vdash \text{ha}^{\tau_1}(\text{app } e_1 \ e_2, \text{app } e'_1 \ e'_2)} \text{ha\_app} \\
\frac{\Delta, x_1^\tau, \text{ha}^{\tau_1}(x, x) \vdash \text{hc}^{\tau_2}(\text{app } e \ x, \text{app } e' \ x)}{\Delta \vdash \text{hc}^{\tau_1 \Rightarrow \tau_2}(e, e')} \text{hc\_arr} \\
\frac{\Delta \vdash \text{wh}^\circ(e_1, e_2) \quad \Delta \vdash \text{hc}^\circ(e_2, e')}{\Delta \vdash \text{hc}^\circ(e_1, e')} \text{hc\_wh} \\
\frac{\Delta \vdash \text{ha}^\circ(e, e')}{\Delta \vdash \text{hc}^\circ(e, e')} \text{hc\_atm}
\end{array}$$

We also give rules for symmetry and transitivity.

$$\begin{array}{c}
\frac{\Delta \vdash \text{hc}^\tau(e', e)}{\Delta \vdash \text{hc}^\tau(e, e')} \text{hc\_sym} \\
\frac{\Delta \vdash \text{hc}^\tau(e_1, e_2) \quad \Delta \vdash \text{hc}^\tau(e_2, e_3)}{\Delta \vdash \text{hc}^\tau(e_1, e_3)} \text{hc\_tr}
\end{array}$$

In LF all of these rules are encoded as follows:

$$\begin{array}{l}
\text{ha\_app: } \text{conc } (\text{ha } E1 \ E1') \rightarrow \text{conc } (\text{hc } E2 \ E2') \\
\quad \rightarrow \text{conc } (\text{ha } (\text{app } E1 \ E2) \ (\text{app } E1' \ E2')) . \\
\text{hc\_arr: } (\{x: \text{tm } T\} \text{hyp } (\text{ha } x \ x) \\
\quad \rightarrow \text{conc } (\text{hc } (\text{app } E \ x) \ (\text{app } E' \ x))) \\
\quad \rightarrow \text{conc } (\text{hc } E \ E') . \\
\text{hc\_wh : } \text{conc } (\text{wh } (E1 : \text{tm } \circ) \ E2) \\
\quad \rightarrow \text{conc } (\text{hc } E2 \ E') \\
\quad \rightarrow \text{conc } (\text{hc } E1 \ E') . \\
\text{hc\_atm: } \text{conc } (\text{ha } (E : \text{tm } \circ) \ E') \\
\quad \rightarrow \text{conc } (\text{hc } E \ E') . \\
\text{hc\_sym: } \text{conc } (\text{hc } E' \ E) \rightarrow \text{conc } (\text{hc } E \ E') . \\
\text{hc\_tr : } \text{conc } (\text{hc } E1 \ E2) \rightarrow \text{conc } (\text{hc } E2 \ E3) \\
\quad \rightarrow \text{conc } (\text{hc } E1 \ E3) .
\end{array}$$

The rules for  $\text{conc}^*$  are defined analogously. The next lemma and theorem illustrate how we can extract the execution trace of the equivalence checking algorithm from a cut-free derivation of  $\text{hc}^\tau(e, e')$ . The formulation of the extraction theorem and its proof follow roughly the same structure as Theorem 4.2.

#### Lemma 5.1 (Congruence)

1. If  $e'_1 \sim e_2 \uparrow \tau$  and  $e_1 \xrightarrow{\text{whr}} e'_1$  then  $e_1 \sim e_2 \uparrow \tau$ .
2. If  $e_1 \sim e_2 \downarrow \tau$  then  $e_1 \xrightarrow{\text{whr}^\downarrow} e'_1$  and  $e_2 \xrightarrow{\text{whr}^\downarrow} e'_2$ .
3. If  $e \sim e' \downarrow \tau$  then  $e' \sim e \downarrow \tau$ .
4. If  $e \sim e' \uparrow \tau$  then  $e' \sim e \uparrow \tau$ .
5. If  $e \sim e' \downarrow \tau$  and  $e' \sim e'' \downarrow \tau$  then  $e \sim e'' \downarrow \tau$ .
6. If  $e \sim e' \uparrow \tau$  and  $e' \sim e'' \uparrow \tau$  then  $e \sim e'' \uparrow \tau$ .

**Proof:** By straightforward structural inductions.  $\square$

**Theorem 5.2 (Extraction)** All derivations in the sequent calculus are assumed to be cut-free. Let

$$\Delta = x_1^{\tau_1}, \text{ha}^{\tau_1}(x_1, x_1), \dots, x_n :^{\tau_n}, \text{ha}^{\tau_n}(x_n, x_n).$$

1. For all derivations of  $\Delta \vdash \text{hc}^\tau(e, e')$  there exists a derivation of

$$\begin{array}{c}
\frac{\quad}{x_1 \sim x_1 \downarrow \tau_1} u_1 \quad \dots \quad \frac{\quad}{x_n \sim x_n \downarrow \tau_n} u_n \\
\vdots \quad \vdots \quad \vdots \\
e \sim e' \uparrow \tau.
\end{array}$$

2. For all derivations of  $\Delta \vdash \text{ha}^\tau(e, e')$  there exists a derivation of

$$\begin{array}{c}
\frac{\quad}{x_1 \sim x_1 \downarrow \tau_1} u_1 \quad \dots \quad \frac{\quad}{x_n \sim x_n \downarrow \tau_n} u_n \\
\vdots \quad \vdots \quad \vdots \\
e \sim e' \downarrow \tau.
\end{array}$$



**Proof:** By induction on the cut-free assertion logic proofs, using Lemma 5.1.

```
ext1 : conc (hc E E')
      -> algC E E' -> type.
ext2 : conc (ha E E')
      -> algA E E' -> type.
```

□

We define the logical relation as a predicate in the assertion logic by instantiating  $P$  (from the general definition of a binary logical relation from Section 2) with  $hc^\circ$ .

### Definition 5.3 (Binary logical relation)

$$\begin{aligned} \llbracket \circ \rrbracket(e, e') &= hc^\circ(e, e') \\ \llbracket \tau_2 \Rightarrow \tau_1 \rrbracket(e, e') &= \forall e_2^{\tau_2}. \forall e_2'^{\tau_2}. \llbracket \tau_2 \rrbracket(e_2, e_2') \\ &\quad \supset \llbracket \tau_1 \rrbracket(\text{app } e \ e_2, \text{app } e' \ e_2') \end{aligned}$$

```
lr : {T:tp} (tm T -> tm T -> form) -> type.
lr_o: lr o hc.
lr_a: lr T2 F -> lr T1 G
      -> lr (T2 => T1) ([e] [e']
                      forall [e2] forall [e2'] (F e2 e2')
                      ==> (G (app e e2) (app e' e2')))
```

## 5.2. Meta-Theory

Next, we extend all theorems from the previous section to the binary logical relation.

**Lemma 5.4 (Escape)** 1. For all types  $\tau$  and assumptions  $\Delta$ , there exists a derivation of  $\Delta \vdash^{\text{cut}} \forall e^\tau. \forall e'^\tau. \llbracket \tau \rrbracket(e, e') \supset hc^\tau(e, e')$ .

2. For all types  $\tau$  and assumptions  $\Delta$ , there exists a derivation of  $\Delta \vdash^{\text{cut}} \forall e^\tau. \forall e'^\tau. \text{ha}^\tau(e, e') \supset \llbracket \tau \rrbracket(e, e')$ .

**Proof:** By induction on  $\tau$ . The proof follows exactly the same structure as the proof of Theorem 4.4, and is hence omitted.

```
esc1: {T:tp} {F:(tm T) -> (tm T) -> form}
      lr T F
      -> (conc* (forall [e] forall [e']
                 F e e' ==> hc e e')) -> type.
esc2: {T:tp} {F:(tm T) -> (tm T) -> form}
      lr T F
      -> (conc* (forall [e] forall [e']
                 ha e e' ==> F e e')) -> type.
```

□

**Lemma 5.5 (Closure under weak head expansion)** For all types  $\tau$  and assumptions  $\Delta$ , there exists a derivation of

$$\begin{aligned} \Delta \vdash^{\text{cut}} \forall e_1^\tau. \forall e_2^\tau. \forall e'^\tau. \\ \text{wh}^\tau(e_1, e_2) \supset \llbracket \tau \rrbracket(e_2, e') \supset \llbracket \tau \rrbracket(e_1, e') \\ \text{and} \\ \Delta \vdash^{\text{cut}} \forall e^\tau. \forall e_1'^\tau. \forall e_2'^\tau. \\ \text{wh}^\tau(e_1', e_2') \supset \llbracket \tau \rrbracket(e, e_2') \supset \llbracket \tau \rrbracket(e, e_1') \end{aligned}$$

in the assertion logic.

**Proof:** By induction on  $\tau$ . Again, the proof follows exactly the same structure as the proof of Lemma 4.5.

```
cwhe: {T:tp} {F:(tm T -> tm T -> form)}
      lr T F
      -> (conc* (forall [e1] forall [e2]
                 forall [e'] wh e1 e2 ==>
                 F e2 e' ==> F e1 e'))
      -> (conc* (forall [e] forall [e1']
                 forall [e2'] wh e1' e2' ==>
                 F e e2' ==> F e e1'))
      -> type.
```

□

**Lemma 5.6 (Symmetry)** For all types  $\tau$ , we show that

$$\Delta \vdash^{\text{cut}} \forall e^\tau. \forall e'^\tau. \llbracket \tau \rrbracket(e, e') \supset \llbracket \tau \rrbracket(e', e).$$

**Proof:** By induction on  $\tau$ , using Def. 5.3, and rules allR, impR, ax, allL, impL, hc\_sym, and cut.

```
sym: lr T F
     -> conc* (forall [e] forall [e']
               F e e' ==> F e' e) -> type.
```

□

**Lemma 5.7 (Transitivity)** For all types  $\tau$ , we show that

$$\Delta \vdash \forall e^\tau. \forall e'^\tau. \forall e''^\tau. \\ \llbracket \tau \rrbracket(e, e') \supset \llbracket \tau \rrbracket(e', e'') \supset \llbracket \tau \rrbracket(e, e'').$$

**Proof:** By induction on  $\tau$  using Def. 5.3, Lemma 5.6, and rules ax, hc\_trans, allR, impR, allL, impL, and cut.

```
tr: lr T F
    -> conc* (forall [e] forall [e']
              forall [e''] F e e' ==>
              F e' e'' ==> F e e'')
    -> type.
```

□

**Theorem 5.8 (Fundamental)** For all types  $\tau$  and assumptions

$$\Delta = x_1^{\tau_1}, y_1^{\tau_1}, \llbracket \tau_1 \rrbracket(x_1, y_1), \dots, x_n^{\tau_n}, y_n^{\tau_n}, \llbracket \tau_n \rrbracket(x_n, y_n)$$

and for all  $e$  with free variables among  $x_1 \dots x_n$  and  $e'$  with free variables among  $y_1 \dots y_n$ . For every derivation of

$$\frac{}{x_1 \equiv y_1 : \tau_1} u_1 \quad \dots \quad \frac{}{x_n \equiv y_n : \tau_n} u_n$$

$$\begin{array}{c} \dots \\ \vdots \\ \dots \end{array}$$

$$e \equiv e' : \tau$$

there exists a derivation of  $\Delta \vdash^{\text{cut}} \llbracket \tau \rrbracket(e, e')$ .

**Proof:** By structural induction on  $e \equiv e' : \tau$  using Def. 5.3, Lemma 5.5, Lemma 5.6, Lemma 5.7, and rules ax, allL, allR, impR, impL, cut and wh\_beta.

```
fund: {E:tm T} {E':tm T} E == E' -> lr T F
      -> conc* (F E E') -> type.
```

□

**Theorem 5.9 (Cut-elimination)** Let  $\Delta$  denote assumptions. Any cut-full sequent derivation of  $\Delta \vdash^{\text{cut}} F$  can be converted into  $\Delta \vdash F$ .

**Proof:** See: Pfenning [Pfe95]. The extra cases introduced by our atomic propositions all fall into the category of “right commutative conversions.”

```
ce : conc* F -> conc F -> type.
```

□

The following theorem summarizes all the work we have done so far. It shows that convertible terms considered equal, can always been recognized as such.

**Theorem 5.10 (Completeness)** For every derivation of  $e \equiv e' : \tau$  there exists a derivation of  $e \sim e' \uparrow \tau$ .

**Proof:**

$e \equiv e' : \tau$	by assumption
$\vdash^{\text{cut}} \llbracket \tau \rrbracket(e, e')$	by Theorem 5.8
$\vdash^{\text{cut}} \forall e^\tau. \forall e'^\tau. \llbracket \tau \rrbracket(e, e') \supset \text{hc}^\tau(e, e')$	by Lemma 5.4
$\vdash^{\text{cut}} \text{hc}^\tau(e, e')$	by ax, allL, impE
$\vdash \text{hc}^\tau(e, e')$	by Theorem 5.9
$e \sim e' \uparrow \tau$	by Theorem 5.2

```
thm: {E:tm T} {E':tm T} eq E E'
     -> algC E E' -> type.
```

□

All proofs presented in this paper are verified by Twelf [SS08].

## 6. Cut-Elimination

Gödel’s second incompleteness theorem tells us that no consistent logic can prove its own consistency, provided the logic is powerful enough to express some very basic coding machinery. Through the lens of the Curry-Howard isomorphism, any typed  $\lambda$ -calculus can be viewed as a logic. Both of the example theorems in this paper, weak normalization and the completeness of a decidable equivalence checking algorithm, imply the consistency of the simply typed  $\lambda$ -calculus when viewed as a logic. This is typical: most theorems that are proven using logical relations are essentially consistency theorems. Thus, it will be impossible to completely formalize any such theorem about a  $\lambda$ -calculus whose proof-theoretic strength exceeds the proof-theoretic strength of the meta-logic.

Recall that, after appealing to the Fundamental Theorem 4.6 and Escape Lemma 4.4 in the proof of Theorem 4.8, we obtain a cut-full derivation of  $\vdash^{\text{cut}} \text{hc}^\tau(e)$ . We could have stopped here if we were certain that this were a meaningful witness to the normalization of  $e$ ; i.e. if we trusted the consistency of the assertion logic. To be sure, we prove cut-elimination and produce a real witness from the cut-free derivation  $\vdash \text{hc}^\tau(e)$ . In general, if the meta-logic is proof-theoretically stronger than the assertion-logic, then it should be possible to prove cut-elimination explicitly.

Our example proofs by structural logical relations reduce the validity of a consistency theorem to the consistency of the assertion logic. Because these reductions use only primitive-recursive machinery, the proof-theoretic strength of the assertion logic must be at least as great as the proof-theoretic strength of the  $\lambda$ -calculus that we are reasoning about. Thus, to scale our examples to Gödel’s T, the assertion logic should provide some notion of induction; to System F, the assertion logic should provide some notion of second-order quantification.

It is worth mentioning that logical consistency is usually proven using either model-theoretic machinery, transfinite induction, or logical relations. Model theoretic machinery is justified by the assumption that set-theory is consistent; transfinite induction is justified by the assumption that a particular ordinal is well founded; a typical logical relations argument is justified by appealing implicitly to the validity of the logical connectives used to define the relation. Structural logical relations bring these assumptions to the surface.

## 7. Conclusion

In this paper we describe the technique of structural logical relations and show how to use it to prove interesting theorems, such as the completeness of the equivalence checking algorithm for the simply typed  $\lambda$ -calculus.

Structural logical relations work well when formalizing proofs by logical relations in proof assistants with a proof-theoretically weak meta-logic but good representational support for working with complex data structures, such as, for example, proof derivations in a sequent calculus.

We do not claim to be the first to characterize a logical relation in terms of the provability of a logical predicate. Indeed, for informal proofs, the idea goes back to Tait's original paper [Tai67]. However, to our knowledge, this is the first work where we separate the roles of the assertion logic and the meta-logic for the purpose of formalization. In proof assistants with strong induction principles, for example Lego or Coq, logical relations are formalized via inductive types [Alt93, BW97, DX06], and reasoned about using elimination forms. Structural logical relations provide not only an alternative, but they make explicit the logical foundations a particular proof is based on.

Proofs by logical relations are popular in large part because they tend to scale well. We believe that our technique preserves this property. For example, in Section 4 and 5 we successfully apply our technique to normalization proofs about the simply typed  $\lambda$ -calculus. Further experiments, such as normalization properties of the monadic  $\lambda$ -calculus (inspired by [LS05]), and about Gödel's T, have shown that the basic structure of the proof remains unchanged but the assertion logic needs to be suitably extended by new inference rules.

Closely related to our work is that of Andreas Abel [Abe08]. He also proves weak normalization directly in Twelf, albeit only for the simply typed  $\lambda$ -calculus. Though he does not use a logical relation, there are striking similarities between the structure of his proof and ours, which should be investigated further.

**Acknowledgments:** We would like to thank Frank Pfenning for pointing out this interesting problem many years ago, and we would also like to thank Andrzej Filinski and Alberto Momigliano for many discussions and comments on earlier drafts of this paper.

## References

- [Abe08] Andreas Abel. Normalization for the simply-typed lambda-calculus in Twelf. In Carsten Schürmann, editor, *Proceedings of the Fourth International Workshop on Logical Frameworks and Meta-Languages*, pages 3 – 16. Elsevier, ENTCS 199, February 2008.
- [ABF<sup>+</sup>05] B. Aydemir, A. Bohannon, M. Fairbairn, J. Foster, B. Pierce, P. Sewell, D. Vytiniotis, G. Washburn, S. Weirich, and S. Zdancewic. Mechanized metatheory for the masses: The POPLmark challenge. In Joe Hurd and Tom Melham, editors, *Proceedings of the Eighteenth International Conference on Theorem Proving in Higher Order Logics*, pages 50–65, Oxford, UK, August 2005. Springer Verlag, LNCS 3603.
- [Alt93] Thorsten Altenkirch. A formalization of the strong normalization proof for System F in LEGO. In M. Bezem and J.F. Groote, editors, *Proceedings of the International Conference on Typed Lambda Calculi and Applications*, pages 13–28, Utrecht, The Netherlands, March 1993. Springer-Verlag LNCS 664.
- [BW97] B. Barras and B. Werner. Coq in Coq. Soumis, 1997.
- [Cra05] Karl Cray. Logical relations and a case study in equivalence checking. In Benjamin C. Pierce, editor, *Advanced Topics in Types and Programming Languages*, pages 223–244. MIT Press, 2005.
- [DX06] Kevin Donnelly and Hongwei Xi. A formalization of strong normalization for simply-typed lambda calculus and System F. In Alberto Momigliano and Brigitte Pientka, editors, *Proceedings of Workshop on Logical Frameworks and Meta-Languages: Theory and Practice*, pages 109–125, Seattle, WA, August 2006. Elsevier, ENTCS 174, Issue 5.
- [Gon05] Georges Gonthier. A computer-checked proof of the four colour theorem. unpublished, 2005.
- [Har96] John Harrison. HOL Light: A tutorial introduction. In Mandayam Srivas and Albert Camilleri, editors, *Proceedings of the First International Conference on Formal Methods in Computer-Aided Design*, pages 265–269, 1996.
- [HHP93] Robert Harper, Furio Honsell, and Gordon Plotkin. A framework for defining logics. *Journal of the Association for Computing Machinery*, 40(1):143–184, January 1993.
- [LCH07] Daniel K. Lee, Karl Cray, and Robert Harper. Towards a mechanized metatheory of standard ML. In *Proceedings of the 34th Annual Symposium on Principles of Programming Languages*, pages 173–184, New York, NY, USA, 2007. ACM Press.
- [LS05] Sam Lindley and Ian Stark. Reducibility and  $\top\top$ -lifting for computation types. In Pawel Urzyczyn, editor, *Proceedings of the*

*Seventh International Conference of Typed Lambda Calculi and Applications*, pages 262–277, Nara, Japan, April 2005. Springer-Verlag, LNCS 3461.

- [MM97] Raymond McDowell and Dale Miller. A logic for reasoning with higher-order abstract syntax: An extended abstract. In Glynn Winskel, editor, *Proceedings of the Twelfth Annual Symposium on Logic in Computer Science*, pages 434–445, Warsaw, Poland, June 1997. IEEE Computer Society Press.
- [Nec97] George C. Necula. Proof-carrying code. In Neil D. Jones, editor, *Proceedings of the 24th Annual Symposium on Principles of Programming Languages*, pages 106–119, Paris, France, January 1997. ACM Press.
- [NU08] Julien Narboux and Christian Urban. Formalising in Nominal Isabelle Crary’s completeness proof for equivalence checking. In Brigitte Pientka and Carsten Schürmann, editors, *Proceedings of the Second International Workshop on Logical Frameworks and Meta-Languages*, pages 3–18. Elsevier, ENTCS 196, January 2008.
- [Pfe92] Frank Pfenning. Computation and deduction. Unpublished lecture notes, 277 pp. Revised May 1994, April 1996, May 1992.
- [Pfe95] Frank Pfenning. Structural cut elimination. In D. Kozen, editor, *Proceedings of the Tenth Annual Symposium on Logic in Computer Science*, pages 156–166, San Diego, California, June 1995. IEEE Computer Society Press.
- [SS08] Jeffrey Sarnat and Carsten Schürmann. Structural logical relations. Project page available at <http://www.twelf.org/slr>, January 2008.
- [Tai67] W. W. Tait. Intensional interpretation of functionals of finite type I. *Journal of Symbolic Logic*, 32:198–212, 1967.